# Brake Specific Fuel Consumption (BSFC)

Brake specific fuel consumption (BSFC) is a parameter that reflects the efficiency of a combustion engine which burns fuel and produces rotational power (at the shaft or crankshaft). In automotive applications, BSFC is used to evaluate the efficiency of the internal combustion engines (ICE). The keyword “brake” is related to the use of a dynamometer (electrical brake) to measure the engine parameters (fuel mass flow rate, torque, etc.).

An internal combustion engine requires fuel and air to produce energy. The amount of fuel used is usually measured on a dynamometer, as a mass flow rate, in kilograms per second [kg/s]. This parameter can not be used to evaluate the efficiency of the engine because it is not obvious how much power we can extract from the fuel. Therefore by dividing the fuel mass flow rate [kg/s] to the engine output power [W], we obtain the brake specific fuel consumption [kg/J]:

$BSFC = \frac{\dot{m}_f}{P_e} \tag{1}$

where:

mf [kg/s] – fuel mass flow rate (measured on the engine dynamometer)
Pe [W] – effective (brake) engine power
BSFC [kg/J] – brake specific fuel consumption

Usually, the fuel mass flow rate is measured in [g/s], the engine power in [kW], which gives the brake specific fuel consumption in [g/kW·h]:

$BSFC = \frac{\dot{m}_f}{P_e} \cdot 3600 \tag{2}$

Engine power is the product between engine speed and torque. Therefore, we can express the BSFC function of engine speed and torque.

$BSFC = \frac{\dot{m}_f}{\omega_e \cdot T_e} \tag{3}$

where:

Te [Nm] – effective (brake) engine torque
ωe [rad/s] – engine speed

Engine torque [Nm] can also be defined function of the mean effective pressure (MEP) of the engine.

$T_e = \frac{n_c V_d p_{me}}{2 \pi n_r} \tag{4}$

where:

nc [-] – number of cylinders
Vd [m3] – cylinder displacement (volume)
pme [Pa] – mean effective pressure
nr [-] – number of crankshaft rotations for a complete engine cycle (for 4-stroke engine nr = 2)

Replacing (4) in (3), we can write the formula of the brake specific fuel consumption function of the mean effective pressure of the engine:

$BSFC = \frac{2 \pi n_r \dot{m}_f}{n_c V_d p_{me} \omega_e} \tag{5}$

The lower the brake specific fuel consumption, the more efficient the engine is. For spark ignition (gasoline engine) the BSFC is around 250 g/kWh and for compression ignition (diesel) around 200 g/kWh.

The brake specific fuel consumption of an engine is usually represented as a contour plot, function of engine speed and torque (or mean effective pressure).

Image: Brake Specific Fuel Consumption (BSFC)

The lowest BSFC is represented by an “island”, usually at mid engine speeds and high torque (load), close to peak full load torque. In the image above we can see that the minimum BSFC is 206 g/kWh at a speed of around 2200 rpm and a mean effective pressure of 15.5 bar.

Image: Brake Specific Fuel Consumption (BSFC) with operating points

Let’s take as example four operating points of the engine (defined by engine speed and MEP). The coordinates of each point and the value of BSFC are centralized in the table below.

 Engine operating point Engine speed [rpm] MEP [bar] (engine load) BSFC [g/kWh] P1 2750 9 240 P2 2750 13 225 P3 3750 13 240 P4 3750 9 260

The most efficient engine operating point is P2, with the BSFC in the area of 240 g/kWh. P1 and P3 have the same brake specific fuel consumption even if the engine is operating at different speed and load. The most inefficient (highest BSFC) operating point is P4.

### Example – How to calculate the BSFC

Let’s take as example a 1.9 liter, 4-stroke spark ignited internal combustion engine, with the following parameters:

• number of crankshaft rotations for a complete engine cycle, nr = 2
• number of cylinders, nc = 4
• cylinder bore, B = 82 mm
• piston stroke, S = 90 mm
• mean effective pressure, pme = 5.16 bar
• engine speed, Ne = 2500 rpm
• fuel mass flow rate, mf = 1.51 g/s

and calculate:

• cylinder capacity (volume), Vd [m3]
• engine brake (effective) torque, Te [Nm]
• engine brake (effective) power, Pe [kW]
• brake specific fuel consumption, BSFC [g/kWh]

Step 1. Calculate the area of the piston

$A_p = \frac{\pi \cdot B^2}{4}=\frac{\pi \cdot 82^2 \cdot 10^{-6}}{4} = 0.005281 \text{ m}^2$

Step 2. Calculate the cylinder capacity (volume)

$V_d = S \cdot A_p = 90 \cdot 10^{-3} \cdot 0.005281 = 0.0004753 \text{ m}^3$

The total engine capacity will be:

$V_t = n_c \cdot V_d = 0.0019012 \text{ m}^3 = 1901.2 \text{ cm}^3 = 1.9012 \text{ L}$

Step 3. Calculate the engine brake (effective) torque

$T_e = \frac{n_c V_d p_{me}}{2 \pi n_r} = \frac{4 \cdot 0.0004753 \cdot 5.16 \cdot 10^5}{2 \cdot \pi \cdot 2} = 78.067 \text{ Nm}$

Step 4. Calculate the engine brake (effective) power

$P_e = N_e \cdot \frac{\pi}{30} \cdot T_e \cdot \frac{1}{1000} = 2500 \cdot \frac{\pi}{30} \cdot 78.067 \cdot \frac{1}{1000} = 20.44 \text{ kW}$

Step 5. Calculate the brake specific fuel consumption

$BSFC = \frac{\dot{m}_f}{P_e} \cdot 3600 = \frac{1.51}{20.44} \cdot 3600 = 265.95 \text{ g/kWh}$

You can also check your results using the calculator below.

### Brake Specific Fuel Consumption (BSFC) Calculator

 nr [-] nc [-] B [mm] S [mm] pme [bar] Ne [rpm] mf [g/s] Engine torque, Te [Nm] = Engine power, Pe [kW] = BSFC [g/kWh] =

### Engine efficiency function of BSFC

Fossil fuel like gasoline and diesel are characterized by a very high energy density. However, due to the combustion process and principle of working of the internal combustion engine, only a fraction of the total fuel energy ends up as mechanical energy at the crankshaft.

Image: Overview of the energy losses in an ICE powered vehicle
Credit: Continental

After combustion, around 30% of the energy is lost through heat, in the engine components. Another 30% of the energy is lost in the exhaust gases. This gives around 40% as indicated energy from which we extract the friction and pumping loses. The overall efficiency of the engine ends up to be around between 25% – 40%.

The efficiency of an internal combustion engine can be calculated function of the brake specific fuel consumption and the lower heating value (LHV) of the fuel.

$\eta_f = \frac{1}{BSFC \cdot Q_{HV}} \tag{6}$

where:

nf [-] – fuel conversion efficiency (engine efficiency)
BSFC [g/kWh] – brake specific fuel consumption
QHV [kWh/g] – fuel lower heating value

For our example, the lower heating value of gasoline is:

$Q_{HV}=18679 \text{ BTU/lb}=43.448 \text{ MJ/kg}=0.012068709 \text{ kWh/g}$

which gives the fuel conversion efficiency:

$\eta_f = \frac{1}{265.95 \cdot 0.012068709} = 0.3115582 = 31.16 \text{ %}$

### Example – BSFC application in Scilab

Using the same engine parameters as in the previous example, let’s calculate the engine brake specific fuel consumption (BSFC) and fuel conversion efficiency maps using fuel mass flow rate measured on the dynamometer, function of the engine speed and torque (mean effective pressure).

Image: Fuel mass flow rate table

The fuel mass flow rate is measured function of the engine torque and engine speed. Since it measures the variation of the fuel mass in time, the fuel mass flow rate is also called the hourly fuel consumption. As expected, the higher the engine speed and load (torque), the higher the fuel mass flow rate. From this perspective, it seems that the most fuel efficient operating points are at low speed and load. This is not the case, since we need to know how much crankshaft power we can extract from the fuel.

If we plot the fuel mass flow rate function of engine speed and torque, we’ll get a surface (map).

Image: Fuel mass flow rate map

Let’s calculate the engine power [kW] from the two axes (engine speed [rpm] and torque [Nm]). To do this, we only need to multiply each engine speed point with each engine torque point, using the formula:

$P_e = T_e \cdot N_e \cdot \frac{\pi}{30} \cdot \frac{1}{1000} \tag{7}$

We should now have the engine power function of the engine speed and load.

Image: Engine power table

If we plot the engine power function of engine speed and torque, we’ll get a surface (map).

Image: Engine power map

Using equation (2) defined above, we can calculate the brake specific fuel consumption of the internal combustion engine. After doing the math, we’ll end up with the following table.

Image: BSFC table

If we plot the brake specific fuel consumption function of engine speed and torque, we’ll get a surface (map).

Image: BSFC map

As you can see, the most efficient engine operating points are between 1500 – 4000 rpm and 90 – 150 Nm. In this engine operating area, the conversion of the energy from fuel to crankshaft is the most efficient.

There are a series of factors which affect the efficiency of the internal combustion engine. The lower the intake air throttling and overall engine losses (friction), the higher the efficiency.

From the same data, we can also generate a contour plot of the BSFC, function of engine speed and torque.

Image: BSFC contour map

Using the lower heating value of gasoline and equation (6), we can calculated the fuel conversion efficiency of the engine. This will show us, in percentage, in which operating points the engine is the most efficient.

Image: Fuel conversion efficiency table

If we plot the brake specific fuel consumption function of engine speed and torque, we’ll get a surface (map).

Image: Fuel conversion efficiency map

Similar to the brake specific fuel consumption, the highest fuel conversion efficiency is obtained at mid engine speeds and high load (torque). From the same data, we can also generate a contour plot of the engine efficiency, function of engine speed and torque.

Image: Fuel conversion efficiency contour map

The Scilab script below captures the calculation of all the tables and generates all the surface and contour plots. The script is split into three parts, the first part contains the input data, the second calculates the engine parameters an the third generates the graphical representations. For an easier understanding of the used equations, each Scilab variable contains in the name the unit of measurement of the parameter (e.g _m for meter, _gpkWh for g/kWh).

// Copyright x-engineer.org
// INPUT DATA
// Lower heating value (gasoline)
Q_HV_kWhpg = 0.012068709;
// Engine parameters
B_m = 82*1e-3; // bore
S_m = 90*1e-3; // stroke
nr = 2; // # of crankshaft revolutions for a complete engine cycle
nc = 4; // # of cylinders
Ne_rpm_y = [500:500:6000]'; // engine speed [rpm] axis
// Engine torque axis
Te_Nm_x = [15.6 31.2 46.8 62.4 78 93.6 109.2 124.8 140.4 156 171.6];
// Fuel mass flow rate
mf_gps_z = [
0.1389 0.2009 0.2524 0.3006 0.3471 0.4264 0.4803 0.5881 0.5881 0.6535 0.7188
0.2777 0.3659 0.4582 0.5587 0.6453 0.7792 0.8977 1.0325 1.1762 1.3069 1.4376
0.4166 0.5538 0.7057 0.8332 0.9557 1.0733 1.2127 1.3428 1.5438 1.9604 2.1564
0.5391 0.7188 0.9116 1.0913 1.2497 1.4115 1.5552 1.7774 2.0290 2.3851 2.8752
0.6330 0.8658 1.0904 1.2906 1.5111 1.6786 1.9440 2.2217 2.4995 2.8997 3.5940
0.7106 0.9949 1.2718 1.5193 1.7888 2.0878 2.3671 2.6661 2.9993 3.5286 4.3128
0.7433 1.0806 1.3722 1.7839 2.2013 2.5490 2.8817 3.1562 3.5507 4.1739 5.0316
0.9475 1.2938 1.7290 2.2087 2.5648 2.9993 3.3391 3.6855 4.2932 4.8355 5.7504
1.1027 1.6026 2.1525 2.5877 2.9957 3.4184 3.8852 4.4108 5.0151 5.6238 6.4692
1.5519 2.0910 2.5730 3.0222 3.4715 3.8717 4.4998 5.0642 5.7781 6.4528 7.1880
1.8868 2.5517 3.1537 3.6479 4.0882 4.4206 5.2203 5.8941 6.5500 7.2329 7.9068
2.0584 2.8817 3.5286 4.0775 4.5578 5.1165 5.6948 6.4300 7.1455 7.8414 8.6256];
//
// PARAMETER CALCULATION
Sp_m2 = %pi*B_m^2/4; // piston area
Vd_m3 = S_m*Sp_m2; // cylinder volume
we_rps_y = Ne_rpm_y * %pi/30; // engine speed [rad/s] axis
// Brake mean effective pressure
BMEP_bar_x = (2*%pi*nr*Te_Nm_x)/(nc*Vd_m3*1e5);
// Engine Power
Pe_kW_z = we_rps_y*Te_Nm_x/1000;
// Brake specific fuel consumption
BSFC_gpkWh_z = (mf_gps_z./Pe_kW_z)*3600;
// Fuel conversion efficiency
etaf_pct_z = 100 ./ (BSFC_gpkWh_z * Q_HV_kWhpg);
//
// ENGINE MAPS (SURFACES)
figure(1)
surf(Te_Nm_x,Ne_rpm_y,Pe_kW_z)
xgrid()
xlabel('Engine torque [Nm]')
ylabel('Engine speed [rpm]')
zlabel('Engine power [kW]')
title('x-engineer.org')
figure(2)
surf(Te_Nm_x,Ne_rpm_y,mf_gps_z)
xgrid()
xlabel('Engine torque [Nm]')
ylabel('Engine speed [rpm]')
zlabel('Fuel mass flow rate [g/s]')
title('x-engineer.org')
figure(3)
surf(Te_Nm_x,Ne_rpm_y,BSFC_gpkWh_z)
xgrid()
xlabel('Engine torque [Nm]')
ylabel('Engine speed [rpm]')
zlabel('Brake Specific Fuel Consumption (BSFC) [g/kWh]')
title('x-engineer.org')
figure(4)
surf(Te_Nm_x,Ne_rpm_y,etaf_pct_z)
xgrid()
xlabel('Engine torque [Nm]')
ylabel('Engine speed [rpm]')
zlabel('Fuel conversion efficiency [%]')
title('x-engineer.org')
figure(5)
contour(Ne_rpm_y,Te_Nm_x,BSFC_gpkWh_z,30)
xgrid()
xlabel('Engine speed [rpm]')
ylabel('Engine torque [Nm]')
title('x-engineer.org')
figure(6)
contour(Ne_rpm_y,Te_Nm_x,etaf_pct_z,30)
xgrid()
xlabel('Engine speed [rpm]')
ylabel('Engine torque [Nm]')
title('x-engineer.org')

### How to improve the BSFC

The internal combustion engine has to generate the power required for propulsion. Depending on the driving scenario and road load, the engine will operate at a certain speed and torque. The same wheel power can be obtained for different values of speed and torque.

Image: The effect of gear ratios on BSFC

As example, the BSFC contour plot above contains also the maximum engine torque curve and the constant engine power curves. By changing the gear we can obtain the same power but at different engine speed and torque.

For example, 100 kW wheel power, can be obtained in 2nd gear, 4th gear and 5th gear. In the 2nd gear, the engine speed is the highest and the engine torque the lowest. Also the brake specific fuel consumption is quite high, around 455 g/kWh.

By shifting up to 5th gear, we maintain the same output power but with higher engine load and lower speed. The advantage is that we’ve put the engine in a more efficient operating point, where the BSFC is around 365 g/kWh.

Vehicles with automatic transmissions have shift schedulers designed to keep the engine in the most efficient operating points through the selected gear. On a manual transmission vehicle is up to the driver to make the right gear shifts in order to minimize the fuel consumption.

Image: BSFC improvement in hybrid electric vehicles

Another way of minimizing the brake specific fuel consumption is by shifting the engine load into a more efficient area while maintaining the same engine speed. This control strategy is only possible in hybrid electric vehicles (HEV), where engine excess torque can be used to generate electrical energy and charge the battery.

In the example above, let’s assume that vehicle is moving at a constant speed, in a fixed gear. The vehicle speed translates in a point of engine torque (100 Nm) and speed (2500 rpm). The engine efficiency in this operating point, 455 g/kWh, is far from optimal. By increasing the engine torque to 250 Nm we decrease the brake specific fuel consumption to 320 g/kWh. The excess torque (150 Nm) will be compensated by the electric machine (-150 Nm) which will run in generator mode and produce electrical energy.

In vehicle with automatic transmissions, the BSFC map is used as a reference to calculate the shift lines for best fuel efficiency. In hybrid electric vehicles, the BSFC map is used in Energy Management strategies to calculate the torque split between engine and electric machine for best fuel efficiency.

For any questions or observations regarding this tutorial please use the comment form below.

Don’t forget to Like, Share and Subscribe!

#### Ad Blocker Detected

Dear user, Our website provides free and high quality content by displaying ads to our visitors. Please support us by disabling your Ad blocker for our site. Thank you!